

Chapter 45 Current Source

The current source is similar to the voltage source. Consider Fig. 45-1.




Fig. 45-1

In Fig. 45-1, there is a current source. Ideally, no matter how R_L changes, the current flowing out of it will remain the same. In practice, there may still be a small change of current. We would like to have a current source in which the amount of the change of current due to the changing of the load is extremely small.

The current source circuits are all based upon the current mirrors. A comprehensive discussion of different current mirrors can be found in Chapter 3. In the following, we will discuss how to obtain a current source which performs very well.

Section 45.1 Current Sources Based upon Elementary

Current Mirrors

Current Source 1

The circuit of Current Source 1 is shown in Fig. 45.1-1.




Fig. 45.1-1 Current Mirror 1

In Fig. 45.1-1, we can see that M1 is now a diode and V_{gs1} is equal to V_{gs2} . Thus, M1 and M2 form a current mirror. The currents $I_{in}(I(R1))$ and $I_{out}(I(R2))$ are determined by V_{D1} , V_{D2} and the IV curve of M2 as shown in Fig. 45.1-2.




Fig. 45.1-2 IV curve, drain voltages and drain currents

I_{out} and I_{in} will be similar to each other if the IV curve is rather flat and V_{D1} is

quite close to V_{D2} . R_L can be changed as shown in Fig. 45.1-1. V_{D1} will not change. V_{D2} will change when R_L is changed. If V_{D2} changes drastically, I_{out} can be quite different from I_{in} as shown in Fig. 45.1-2. The following experiment will show the performance of Current Source 1. Throughout this chapter, when we mention error of a current source, we mean the difference between I_{out} and I_{in} divided by I_{in} .

Experiment 45.1- 1 The Testing of Current Source 1




Fig.45.1-3 A current mirror with an elementary current mirror

In this experiment, Current Source 1 is tested. For the convenience of discussion, we display the circuit of Current Source 1 in Fig. 45.1-3. The program is in Table 45.1-1 and the results are in Fig. 45.1-4.

Table 45.1-1 The program of Experiment 45.1-1

```
.protect
.lib "C:\mm0355v.l" TT
.unprotect
.op
.option post

M1 iin iin 0 0 nch w=10u l=1u
M2 iout iin 0 0 nch w=10u l=1u

.param R_val=1K
```


```

Vdd Vdd 0 3.3
*i1 Vdd iin 10u
R1 Vdd iin 3K
RL Vdd iout 'R_val'


.dc R_val 1K 10K 10
.probe i(R1) i(RL) par('abs(i(RL)- i(R1))*100/i(R1)')

.end


```




(a)



(b)



(c)



(d)

Fig. 45.1-4 The results of Experiment 45.1-1

The results of Experiment 45.1-1 are listed as follows.

- (1) From Fig. 45.1-4(a), we can see that I_{out} is similar to I_{in} only for a rather short range of R_L .
- (2) From Fig. 45.1-4(b), we can see that V_{D2} changes quite a lot while V_{D1} is kept a constant. As shown in Fig. 45.1-2, this will cause an error.
- (3) From Fig. 45.1-4(c), we can see that the error reaches 10% when R_L is 5K and it reaches 50% when R_L is 10K.
- (4) From Fig. 45.1-4(d), we can see that the IV curve of M2 is not flat enough.

Current Mirror 2

From the above discussion, we understand that the current source would perform better if the IV curve is flatter. The IV curve will be flatter if we increase the sizes of M1 and M2. The circuit of Current Source 2 is the same as that of Current Source 1. But the sizes of M1 and M2 are enlarged. The next experiment shows the performance of Current Source 2.

Experiment 45.1-2 The Testing of Current Source 2

In this experiment, we increase the W of M1 and M2 from 10u to 30u and the L from 1u to 3u. The program is in Table 45.1-2 and the results are in Fig. 45.1-5.

Table 45.1-2 The program of Experiment 45.1-2

```
current source
.protect
.lib "C:\mm0355v.l" TT
.unprotect
.op
.option post


M1 iin iin 0 0 nch w=30u l=10u
M2 iout iin 0 0 nch w=30u l=10u

.param R_val=1K

Vdd Vdd 0 3.3
*i1 Vdd iin 10u
R1 Vdd iin 3K
RL Vdd iout 'R_val'

.dc R_val 1K 10K 10
.probe i(R1) i(RL) par('abs(i(RL)- i(R1))*100/i(R1)')

.end
```




(a)


(b)



(b)



(c)



(d)

Fig. 45.1-5 The results of Experiment 45.1-2

From 45.1-5, we can see that Current Source 2 is better than Current Source 1. This is because its IV curve is flatter as shown in Fig. 45.1-5(d),

Experiment 45.1-3 The Changing of $I_{in}(I(R1))$

In this experiment, R1 is increased from 3K to 6K. The program is in Table 45.1-3 and the result is in Fig. 45.1-6. $I_{in}(I(RL))$ decreases from 455uA to 282uA.

Table 45.1-3 The program of Experiment 45.1-3

```

current source
.protect
.lib "C:\mm0355v.l" TT
.unprotect
.op
.option post

M1 iin iin 0 0 nch w=30u l=10u
M2 iout iin 0 0 nch w=30u l=10u

.param R_val=1K

Vdd Vdd 0 3.3
*i1 Vdd iin 10u
R1 Vdd iin 6K
RL Vdd iout 'R_val'

.dc R_val 1K 10K 10
.probe i(R1) i(RL) par('abs(i(RL)- i(R1))*100/i(R1)')

.end

```




Fig. 45.1-6 The result of Experiment 45.1-3

In this experiment, we test Current Source 3 whose circuit is shown in Fig. 45.2-1. The program is in Table 45.2-1 and the results are in Fig. 45.2-2.

Table 45.2-1 The program of Experiment 45.2-1

```
current mirror
.protect
.lib "C:\mm0355v.l" TT
.unprotect
.op
.option post

M1 iin 1 0 0 nch w=30u l=10u
M2 1 1 0 0 nch w=30u l=10u
M3 iout iin 1 0 nch w=30u l=10u

.param R_val=1K

Vdd Vdd 0 3.3
*i1 Vdd iin 10u
R1 Vdd iin 3K
RL Vdd iout 'R_val'

.dc R_val 1K 10K 10
.probe i(RL) par('abs(i(RL)- i(R1))*100/i(R1)')

.end
```





Fig. 45.2-2 The results of Experiment 45.2-1

Let us review Fig. 45.2-2.

- (1) From Fig. 45.2-2(a), we can see that $I_{out}(I_{RL})$ is close to $I_{in}(I(R1))$ for a wide range of R_L .
- (2) From Fig. 45.2-2(b), we can see that V_{D2} almost does not change. Note that for the current sources introduced in the above section, V_{D2} changes quite a lot.
- (3) From Fig. 45.2-2(c), we can see that the error of Current Source 3 is reduced to 0.59% in the worst case while that of Current Source 2 is 38%.

We may conclude that Current Source 3 is much better than the other two current sources introduced in Section 45.1.

Current Source 4 An Improved Version of Current Source 3

The circuit of Current Source 4 is shown Fig. 45.2-3.




Fig. 45.2-3 Current Source 4

It is easy to see that Current Source 4 is more balanced than Current Source 3. As shown by the following experiment, this will make V_{D2} be closer to V_{D1} and the error further smaller.

Experiment 45.2-2 The Testing of Current Source 4

In this experiment, we test Current Source 4. The program is in Table 45.2-1 and the results are in Fig. 45.2-4.

Table 45.2-2 The program of Experiment 45.2-2

```
current mirror
.protect
.lib "C:\mm0355v.l" TT
.unprotect
.op
.option post

M1 2 1 0 0 nch w=30u l=10u
M2 1 1 0 0 nch w=30u l=10u
M3 iout iin 1 0 nch w=30u l=10u
M4 iin iin 2 0 nch w=30u l=10u

.param R_val=1K

Vdd Vdd 0 3.3
*i1 Vdd iin 10u
R1 Vdd iin 3K
RL Vdd iout 'R_val'

.dc R_val 1K 10K 10
.probe i(RL) par('abs(i(RL)- i(R1))*100/i(R1)')

.end
```






Fig. 45.2-4 The results of Experiment 45.2-2

From Fig. 45.2-4, we can see that V_{D2} is equal to V_{D1} . This makes the error to be as small as 0.007%

In conclusion, we succeeded in obtaining a current source whose error is around 1/10000 of the input current. That is, our current source is really a constant current source which does not change much as its load changes.